
付録１ 色付きxyplotグラフ(show3の使用方法)

注意
　このサイト「モデル記述言語CASTによるシミュレーション」[注2]で公開されているsimcast.zip（＋extProlog）をインストールすると，モデル理論アプローチによるシミュレーション開発実行環境simcastができあがり，本文中のモデルを実行することができる．開発実行環境simcastの操作方法などは，同サイトや教科書[注3]を参照されたい．

１ はじめに

　本稿では，技術資料[注1]にもとづいて，モデル理論アプローチによるシミュレーション開発実行環境Simcastにおいて，簡便にグラフ表示を行なうことができる述語show1, show2, show3の使用方法について述べる．これらをshow系の述語と呼ぶことにする．（show系以外にも，座標軸を表示しないwopen系のグラフ表示の方法があるが，それについては別の技術資料[注5]を参照されたい．）

　show系の述語は次の3つの特徴を持っている．

1）グラフウインドウを開くための事前操作が不要である．

2）グラフの種類に合わせた座標軸が表示される．

3）数値の大小に応じて座標軸のスケール（刻み幅）が変化する．

スケールの変化については体験してもらいたい．

　本稿では主として述語の使い方について述べる．show1とshow2の述語の代表的な書式は次のようになる（show3については付録を参照せよ）．

【書式】

(1) show2(番号,データ, グラフの種類)
(2) show1(データ, グラフの種類)

【例】

(1) show2(3,[10, 20,30],"plot")

(2) show1([10,20,30],"plot")

【意味】

(1) out_3.gという名前のグラフウインドウを開き，座標平面上に３点(0,10),(1,20),(2,30)を結んだ折れ線を描く． 

(2) out_0.gという名前のグラフウインドウを開き，折れ線を描く．show2(0, [10,20,30],"plot")と同じ．

　show2は複数のグラフウインドウを番号つきの名称（out_番号.g）で区別しているが，show1ではひとつのグラフウインドウout_0.gだけを使う[注4]．以下では，各種のグラフを描くオートマトンモデルを例示する．
（表紙の注意）

２　plotグラフ（折れ線）
　座標平面に点を打って，直線でつないでゆく（折れ線を描く）．
例

　show1([20,30,50], "plot"),　　　// ３点(0,20),(1,30),(2,50)を結ぶ折れ線をひとつ描く．

　show1([[1,2,3],[5,3,4]], "plot"),　// ２つの折れ線を描く．

演習問題

　乱数を100個発生させて，折れ線を作れ．Hint：オートマトンの状態遷移のときに乱数を発生させ，それを次の状態とせよ．出力関数にグラフ表示を定義せよ．ダミー入力を数100回入力せよ．

図表1　乱数をグラフ表示する

ヒント：関数myrandom( )で0以上1以下の乱数を発生させることができる．次のモデルはDialogウインドウに乱数を数値として表示していくものである．

inputsequence()="d";

times()=100;

initialstate()=c0 <-> c0:=0;
delta(c,a)=cc <-> cc:= myrandom()-0.5;
このモデルでは，状態cは -0.5から +0.5までの間の実数となる．演習問題の主旨は，状態cの列を蓄積しながらグラフを表示せよ，ということである．

（答）

//show01.set

inputsequence()="d";

times()=100;

initialstate()=c0 <-> c0:=0, hist.g:=[];
delta(c,a)=cc <-> cc:= myrandom()-0.5;
lambda(c)=b <-> b:=c,


hist:=hist.g,


hist1:=append(hist,[c]),

show1(hist1,"plot"),

hist.g:=hist1;

　図表2　乱数をグラフ表示するユーザモデル

　ここでは，初期状態initialstateの設定のときに，グローバル変数hist.gを用意しておき，そこに状態の履歴を保存するようにした．また，出力関数lambdaがグラフ表示を行なうことにした．発生した乱数cはhist1:=append(hist,[c])で変数hist1に蓄積され，show1で表示されている．結果として，履歴はhist.g:=[ 0, 0.32, -0.41, ... ]のようなリストになっていく．それをshow1でグラフ表示している．強制的に時間を進めるため，inputsequence()="d"; times()=100; で，（状態遷移に影響しない）ダミーの"d"を100回入力している．

　しかし，入力回数をtimes()=1000に変更したい（1000回処理を行ないたい）場合は，ひとつの画面に多数の点を表示すると，見て理解することが困難になるので，適当に削りたい．
演習問題
　乱数を1000個発生させて，折れ線を作れ．ただし，常に最新の200データだけを表示せよ．Hint：リストzのi番目からj番目までを抽出する述語project(z,["r", i, j])を使え，

（答）

//show01a.set

inputsequence()="d";

times()=1000;

initialstate()=c0 <-> c0:=0, hist.g:=[];

delta(c,a)=cc <-> cc:=myrandom()-0.5;
lambda(c)=b <-> b:=c,


hist:=hist.g,


hist1:=append(hist,[c]),

len:=cardinality(hist1),


hist2:=project(hist1,["r",len-199,len]),
show1(hist2,"plot"),

hist.g:=hist2;

　上記モデルでは，len:=cardinality(hist1)で履歴の長さを求め，project(hist1,["r",len-199,len])で，最新200回分のデータを抽出している．len-199 < 0 のときは，0からlenまでの要素が抽出されるので問題ない．
演習問題

　関数sin(θ)のグラフを描け．ただし，刻み幅は2π/100ラジアンとする．また，常に最新の200データを表示せよ．

[image: image3.png]



（答）　数値としてsin(θ)を計算するだけなら，次のようになる．

//show04.set
inputsequence()="d";

times()=1000;

initialstate()=c0 <-> c0:=0;

delta(c,a)=cc<-> cc:=c+2*.pi/100;

lambda(c)=b <-> b:=sin(c);

　このモデルでは，状態cは角度θに相当する．状態遷移関数deltaは変化していくθの数値を求め，出力関数lambdaの中のsin(c)で，sin(θ)をグラフ表示している．ただし，.piは定項で，.pi = 3.14159284 である．

　図形として描画するには，ウインドウを開き，そこにデータを表示する必要がある．ここでは述語show1を利用する．


//show05.set

inputsequence()="d";

times()=1000;

initialstate()=c0 <-> c0:=0, hist.g:=[];

delta(c,a)=cc<-> cc:=c+2*.pi/100;
lambda(c)=b <-> b:= sin(c),


hist:=hist.g,


hist1:=append(hist,[b]),

len:=cardinality(hist1),


hist2:=project(hist1,["r",len-199,len]),

show1(hist2,"plot"),

hist.g:=hist2;
　例えば，述語show1(hist2,"plot")を使わず，show2(0,hist2,"bar"), show2(1,hist2,"plot")に変更すると，２つのグラフウインドウが開き，同時に２つのグラフが変化していく（読者はそれを試みよ）． 

３　xyplotグラフ

　xy平面座標の上に，点(x,y)を打ってゆく．

書式　show1([xList,yList], "xyplot")

//データ=[x座標のリスト, y座標のリスト]
例　　show1([[3,4,5],[10,20,30]], "xyplot")
//３点(3,10), (4,20), (5,30)を打つ．
演習問題

　２つづつ乱数を発生させて，平面上に1000個の点(c1,c2)を打て．ただし常に最新の200個を表示せよ．
[image: image4.png]



（答）


//show02.set

inputsequence()="d";

times()=1000;

initialstate()=c0 <-> c0:=[0.5,-0.5], hist.g:=[];

delta([x,y],a)=[xx,yy] <-> [xx,yy]:= [myrandom()-0.5, myrandom()-0.5];
lambda(c)=b <-> b:=c,

hist:=hist.g,


hist1:=append(hist,[c]),

len:=cardinality(hist1),


hist2:=project(hist1,["r",len-199,len]),


Ds:=transpose(hist2),
show1(Ds,"xyplot"),

hist.g:=hist2;

　上記モデルの状態遷移関数deltaの定義は1行で書かれており，2変数[xx, yy]に乱数を「同時に」代入している．2行で書くなら，
x2:= myrandom()-0.5,

y2:= myrandom()-0.5;
となり，同じ意味である．

　出力関数lambdaでグラフ表示を行なうが，履歴はhist.g=[[x1,y1],[x2,y2],[x3,y3],......]という形で蓄積されていく．ここでは，リストのリストが「行列」の表現でもあることを考慮し，転置行列を求める関数tranposeを使ってDs = [ [x1,x2,x3,...], [y1,y2,y3,...] ] の形に変換している．本稿では，この方法を推奨する．

４　trajectoryグラフ


show1(データ,"trajectory")とすると，xyplotした各点がつながれてゆく．
例　show1([[3,4,5],[10,20,30]], "trajectory"),　　// ３点(3,10), (4,20), (5,30)を結ぶ．

（答）


//show02c.set

inputsequence()="d";

times()=1000;

initialstate()=c0 <-> c0:=[0.5,-0.5], hist.g:=[];

delta([x,y],a)=[xx,yy] <-> [xx,yy]:= [myrandom()-0.5, myrandom()-0.5];
lambda(c)=b <-> b:=c,

hist:=hist.g,


hist1:=append(hist,[c]),

len:=cardinality(hist1),


hist2:=project(hist1,["r",len-199,len]),


Ds:=transpose(hist2),
show1(Ds,"trajectory"),

hist.g:=hist2;

５ おわりに

　本稿では，技術資料[1]を参考にして，グラフがとにかく描けることを報告した．同資料によれば，グラフの種類には，円グラフ（"pie"），棒グラフ（"bar"），レーダグラフ（"radar"），折れ線グラフ（"plot"），平面座標上の点（"xyplot"），軌跡の表示（"trajectory"）がある．

　円グラフのデータは，総和が１であること，また，レーダグラフのデータは正の数でなくてはならない．一つのウインドウに複数の折れ線（"plot"）を描けるが，線の数に限界があるらしい（未確認）．
　1000回くらいオートマトンが状態遷移を行なうと，パソコンのCPUが発熱し，換気扇が回り始める．ウインドウの数が増えると処理速度は遅くなる．グラフ表示の処理負荷は相当に高い．
注

[1] 高原康彦（代表）「マニュアル第２版」平成13-14年度科学研究費報告書

　http://www.geocities.jp/kisoten/background/（2008.04.11）

述語show1, show2はMTA-SDKのdssディレクトリでも使用可能である．

[2] 旭貴朗「モデル記述言語CASTによるシミュレーション」
　http://www2.toyo.ac.jp/~asahi/research/simulation/（2008.02.23）

[3] 高原康彦ほか『形式手法 モデル理論アプローチ：情報システム開発の基礎』日科技連(2007) の少なくとも第３章3.1項pp.87-103 を参照されたい．

　http://www.geocities.jp/kisoten/（2008.04.11）

[4] ウインドウの種類は，グラフ，スプレッドシート，テキストがある．すべてのウインドウはウインドウ番号Wpとウインドウ名称Wnameをもつ．show系のグラフ表示の場合にはウインドウ名称（out_n.g）で区別する．（一方，wopen系のグラフ表示の場合は，ウインドウ番号で区別する．）末尾が.gではあるがグローバル変数とは無関係のようである．一方，スプレッドシートの場合はウインドウ番号で区別する．
[5] 技術資料「座標軸の非表示と多数データの描画」所在はサイト[2]．

付録１　色付きxyplotグラフ(show3の使用方法)
　まだ制限が多く，洗練されていないが，座標(i,j)に色付きの点を描くことができる．色は数値vで５種類を指定することができる．

【書式】show3(n, [IList, JList, VList], "xyplot","color")
　ただし，nはウインドウ名称（out_n.g）を指定する自然数である．また，IListとJListはそれぞれ自然数を要素とするリストであり，VListは（色を指定するための）0以上の実数を要素とするリストである．３つのリストIList, JList, VListは，要素数が同じでなければならない．色指定は５種類あり，1 (orange), 2 (green), 3 (purple), 4 (white), 5 (red)である．これ以上の数値に対しても同じ順序で繰り返される．したがって，v=6のときはoranegeで，v=10のときはredである．特別にv=0だけは無視される（あるいは背景色と同じになる）ことに注意する．この述語show3は，既存のshow1(L,Gr,"color")を筆者が修正したものである．

【例１】

//showColor00.set

inputsequence()=As <-> As:=[0];

initialstate()=c <-> c:=0;

delta(c,a)=cc <-> cc:=c,

Data:=[[1,1,1],[2,2,2],[3,3,3],[4,4,4],[5,5,5]],


Ds:=transpose(Data), 


show3(2,Ds,"xyplot","color");
【意味】out_2.gという名前のグラフウインドウを開き，座標平面上に５点(1,1), (2,2), (3,3), (4,4), (5,5)を描き，それぞれの点に数値1, 2, 3, 4, 5に対応する色をつける．

　この例では，観測データData:=[[1,1,1],[2,2,2],[3,3,3],[4,4,4],[5,5,5]]をグラフ表示するために，まず転置（transpose）をしている．したがって，Ds=[IList, JList, VList]は，実際には，

IList=[1,2,3,4,5], JList=[1,2,3,4,5], VList=[1,2,3,4,5] となっている．[i,j,v]の形をしたデータのリストは，転置をしてからshow3に代入することがポイントである．

【例２】

//showColor08.set
Ps.g=product(genIndex(1, 30),genIndex(1, 30));

inputsequence()=As <-> As:=genIndex(1,100);

initialstate()=c <-> c:=[[3,3,3]];

delta(c,a)=cc <-> cc:=defSet( p(z,[i,j],[a]),member([i,j],Ps.g) );

p(z,[i,j],[a]) <-> z:=[i, j, a+i];
lambda(c)=b <-> 
b:=project(c,0),



L:=transpose(c), show3(2,L,"xyplot","color");

　この例では，まず座標の範囲Ps.g={(i,j)｜i∈{1,2,...,30}, j∈{1,2,...,30}}を指定している．つまり一度に30×30 = 900個の点に対して色をつけることができる．入力aが引き起こす次状態は述語defSetで定義されており，cc={(i,j,v)｜v = a+i , (i,j)∈Ps.g}である．したがって，点 (i,j) の色はv=a+iとなる．入力はa=1,2,3,...,100 と変化するので，グラフは動的に変化していき，右から左へ縞模様が流れてゆくことになる．

【制限】　表示範囲については，Mac OS Xの場合，30×30 = 900個までは正常に表示された．これ以上になると，縦方向に関して間引き現象が起こる．40×40のときは開発環境がダウンした．また，常に縦軸はjを示さず，vの最大値を示すので，縦軸の目盛は無視せざるを得ない．なぜか「ディレクトリusr/local/ooparaを作成せよ」との警告が出たことを記しておく．これらの制限の原因は究明していない．

【演習】

　上の例２の述語pを変更し，

p(v,[i,j],[a]) <-> v:= [i,j,a/2 +(i*i+j*j)*7/2000]; 

にすると，右のような図ができ，動的に変化していく．

やってみよう．

